```
> restart:
```

Musterlösung Aufgabe 8

Unbestimmte Integration rationaler Funktionen, Bernoulli-Darstellung

Gegeben sei die rationale Funktion $r(x) = \dfrac{x^{10} + 2\,x^7 - 5\,x^5 - x + 1}{x^8 + x^7 - 2\,x^6 - x^5 + 2\,x^4 - x^3 - 2\,x^2 + x + 1}$ **.**

a) Berechnen Sie das unbestimmte Integral in der Bernoulli-Darstellung, d.h. zeigen Sie, dass sich die Stammfunktion in der Form

$$\int r(x)\,dx \;=\; \int P(x)\,dx \quad + \sum_{i=1}^{M} \frac{p_i(x)}{\left(x - a_i\right)^{\nu_i - 1}} + \sum_{i=1}^{M} A_i \cdot \log\left(x - a_i\right)$$

schreiben lässt. Hierbei sind die $a_i \in \mathbb{C}, i = 1, ..., M,$ **die verschiedenen Nullstellen des Nennerpolynoms mit den jeweiligen Vielfachheiten** ν_i **und** $P(x), p_i(x) \in \mathbb{C}[x]$ **Polynome mit** $deg(P(x)) \le deg(r(x)) = deg(p(x)) - deg(q(x))$ **sowie** $deg\left(p_i(x)\right) < \nu_i - 1$**.**

Hinweis: Nutzen Sie convert,fullparfrac für die Zerlegung des Nennerpolynoms in Linearfaktoren.

```
> r:=x->(x^10+2*x^7-5*x^5-x+1)/(x^8+x^7-2*x^6-x^5+2*x^4-x^3-2*x^2+
  x+1);
```

$$r := x \mapsto \frac{x^{10} + 2 \cdot x^7 - 5 \cdot x^5 - x + 1}{x^8 + x^7 - 2 \cdot x^6 - x^5 + 2 \cdot x^4 - x^3 - 2 \cdot x^2 + x + 1} \quad (1)$$

```
> convert(r(x),fullparfrac);
```

$$x^2 - x + 3 + \left(\sum_{\_\alpha = RootOf(\_Z^2 - \_Z + 1)} \frac{\frac{\_\alpha}{9} + \frac{2}{9}}{x - \_\alpha}\right) - \frac{1}{8\,(x-1)^2} + \frac{1}{4\,(x-1)} - \frac{101}{36\,(x+1)} - \frac{5}{6\,(x+1)^3} + \frac{1}{2\,(x+1)^4} + \frac{47}{24\,(x+1)^2} \quad (2)$$

```
> int((2),x);
```

$$3\,x + \frac{x^3}{3} - \frac{x^2}{2} + \frac{1}{8\,(x-1)} + \frac{\ln(x-1)}{4} - \frac{1}{6\,(x+1)^3} + \frac{5}{12\,(x+1)^2} - \frac{47}{24\,(x+1)} - \frac{101\,\ln(x+1)}{36} + \sum_{\_\alpha = RootOf(\_Z^2 - \_Z + 1)} \left(\frac{\ln(x - \_\alpha)\,\_\alpha}{9} + \frac{2\,\ln(x - \_\alpha)}{9}\right) \quad (3)$$

```
> convert((3),radical);
```

(4)

$$3\,x+\frac{x^3}{3}-\frac{x^2}{2}+\frac{1}{8\,(x-1)}+\frac{\ln(x-1)}{4}-\frac{1}{6\,(x+1)^3}+\frac{5}{12\,(x+1)^2}-\frac{47}{24\,(x+1)}$$
$$-\frac{101\,\ln(x+1)}{36}+\frac{\ln\left(x-\frac{1}{2}-\frac{\mathrm{I}\sqrt{3}}{2}\right)\left(\frac{1}{2}+\frac{\mathrm{I}\sqrt{3}}{2}\right)}{9}+\frac{2\,\ln\left(x-\frac{1}{2}-\frac{\mathrm{I}\sqrt{3}}{2}\right)}{9}$$
$$+\frac{\ln\left(x-\frac{1}{2}+\frac{\mathrm{I}\sqrt{3}}{2}\right)\left(\frac{1}{2}-\frac{\mathrm{I}\sqrt{3}}{2}\right)}{9}+\frac{2\,\ln\left(x-\frac{1}{2}+\frac{\mathrm{I}\sqrt{3}}{2}\right)}{9} \quad \textbf{(4)}$$

```
> F1:=unapply( collect((4), ln) ,x);   # Zusammenfassen der logarithmischen Ateile.
```

$$F1 := x \mapsto \frac{\ln(x-1)}{4}-\frac{101\cdot\ln(x+1)}{36}+\left(\frac{5}{18}+\frac{\mathrm{I}\cdot\sqrt{3}}{18}\right)\cdot\ln\left(x-\frac{1}{2}-\frac{\mathrm{I}\cdot\sqrt{3}}{2}\right)+\left(\frac{5}{18}-\frac{\mathrm{I}\cdot\sqrt{3}}{18}\right)\cdot\ln\left(x-\frac{1}{2}+\frac{\mathrm{I}\cdot\sqrt{3}}{2}\right)+3\cdot x+\frac{x^3}{3}-\frac{x^2}{2}+\frac{1}{8\cdot(x-1)}-\frac{1}{6\cdot(x+1)^3}+\frac{5}{12\cdot(x+1)^2}-\frac{47}{24\cdot(x+1)} \quad \textbf{(5)}$$

b) Zeigen Sie, dass sich die Lösung des Maple-Integrators `int` von der Bernoulli-Darstellung um eine Konstante unterscheidet. Ermitteln Sie diese Konstante und plotten Sie die Differenz der Lösungen.

```
> F2:=unapply(int(r(x),x),x);
```

$$F2 := x \mapsto \frac{x^3}{3}-\frac{x^2}{2}+3\cdot x-\frac{1}{6\cdot(x+1)^3}+\frac{5}{12\cdot(x+1)^2}-\frac{47}{24\cdot(x+1)}-\frac{101\cdot\ln(x+1)}{36}+\frac{1}{8\cdot(x-1)}+\frac{\ln(x-1)}{4}+\frac{5\cdot\ln(x^2-x+1)}{18}-\frac{\sqrt{3}\cdot\arctan\left(\frac{(2\cdot x-1)\cdot\sqrt{3}}{3}\right)}{9} \quad \textbf{(6)}$$

```
> diff(F1(x)-F2(x),x);
```

$$\frac{\frac{5}{18}+\frac{\mathrm{I}\sqrt{3}}{18}}{x-\frac{1}{2}-\frac{\mathrm{I}\sqrt{3}}{2}}+\frac{\frac{5}{18}-\frac{\mathrm{I}\sqrt{3}}{18}}{x-\frac{1}{2}+\frac{\mathrm{I}\sqrt{3}}{2}}-\frac{5\,(2\,x-1)}{18\,(x^2-x+1)}+\frac{2}{9\left(\frac{(2\,x-1)^2}{3}+1\right)} \quad \textbf{(7)}$$

```
> simplify((7));
```

$$0 \quad \textbf{(8)}$$

Die Ableitung der Differenz von $F1$ und $F2$ verschwindet, also unterscheiden sich die Stammfunktionen nur um eine Konstante. Durch Einsetzen eines speziellen Wertes in den Differenzausdruck erhalten wir den Wert der Konstanten.

```
> subs(x=0,F1(x)-F2(x));
```

$$\left(\frac{5}{18}+\frac{I\sqrt{3}}{18}\right)\ln\left(-\frac{1}{2}-\frac{I\sqrt{3}}{2}\right)+\left(\frac{5}{18}-\frac{I\sqrt{3}}{18}\right)\ln\left(-\frac{1}{2}+\frac{I\sqrt{3}}{2}\right)-\frac{5\ln(1)}{18}+\frac{\sqrt{3}\arctan\left(-\frac{\sqrt{3}}{3}\right)}{9} \qquad \textbf{(9)}$$

```
> simplify((9));
```

$$\frac{\sqrt{3}\,\pi}{18} \qquad \textbf{(10)}$$

```
> evalf((10));
```

$$0.3022998942 \qquad \textbf{(11)}$$

Plot der Differenz von $F1$ **und** $F2$.

```
> plot(F1(x)-F2(x),x=-10..10);
```




```
> plot(F1(x)-F2(x),x=-10..10,0.3..0.31);
```