```
> restart:
> interface(imaginaryunit=i):
```

Musterlösung Aufgabe 2

Erweitertes Ersatzschaltbild (Eindiodenmodell) der Solarzelle

Eine Solarzelle hat physikalisch den gleichen Aufbau wie eine Diode. Sie besteht aus einem n- und p-dotierten Halbleiter mit einer sich ausbildenden Raumladungszone, sodass sich eine unbestrahlte Solarzelle wie eine Diode verhält und sich im einfachsten Fall durch eine Diode beschreiben lässt. Im erweiterten Ersatzschaltbild der Solarzelle sind Zellstrom I und Zellspannung U durch die folgende implizite Gleichung miteinander verknüpft:

```
> I=I[ph]-I[s]*(exp((U+I*R[s])/(m*U[T]))-1)-(U+I*R[s])/R[p];
```

$$I = I_{ph} - I_s \left(e^{\frac{I R_s + U}{m U_T}} - 1 \right) - \frac{I R_s + U}{R_p} \tag{1}$$

Hierbei bedeuten I_{ph} Photostrom, I_s Sättigungsstrom in Diodensperrrichtung, U_T Temperaturspannung, R_s Serienwiderstand, R_p Parallelwiderstand und m Diodenfaktor.

a) Lösen Sie die Gleichung (1) symbolisch mit Hilfe der LambertW-Funktion nach I.

```
> I=solve((1),I);
```

$$I = -\frac{1}{R_s} \left(-\left(-\mathrm{LambertW}\left(-\frac{I_s R_p R_s \, e^{\frac{R_p \left(I_{ph} R_s + I_s R_s + U\right)}{m U_T \left(R_p + R_s\right)}}}{-m R_p U_T - m R_s U_T} \right) + \frac{R_p \left(I_{ph} R_s + I_s R_s + U\right)}{m U_T \left(R_p + R_s\right)} \right) m U_T + U \right) \tag{2}$$

b) Gegeben seien die Zellparameter:
$m = 1.0$, $U_T = 0.025$, $I_s = 1.000000000\ 10^{-10}$, $R_p = 10.0$, $I_{ph} = 3.0$ **. Zeichnen Sie die I-U-Kennlinien (Zellstrom I in Abhängigkeit von der Zellspannung U) für die Serienwiderstände** $R_s = 0.001, 0.01, 0.05, 0.1, 0.2$ **in ein Diagramm über dem Intervall** $U = 0\,..\,0.6$ **.**

Hinweis: Erstellen Sie für den Plot eine Liste mit Ausdrücken, z.B. mit subs und map.

```
> subs([m=1.0,U[T]=0.25e-1,I[s]=1.0*10^(-10),R[p]=10.0,I[ph]=3.0],
```

```
rhs((2)));
```

$$-\frac{1}{R_s}\left(0.0250\,\mathrm{LambertW}\left(-\frac{1.000000000\times 10^{-9}\,R_s\,\mathrm{e}^{\frac{400.0000000\,(U+3.000000000\,R_s)}{10.0+R_s}}}{-0.0250\,R_s-0.25000}\right)-\frac{10.00000000\,(U+3.000000000\,R_s)}{10.0+R_s}+U\right) \quad \textbf{(3)}$$

```
> Rs:=[0.001,0.01,0.05,0.1,0.2];
```

$$Rs := [0.001, 0.01, 0.05, 0.1, 0.2] \quad \textbf{(4)}$$

```
> p:=map(x->subs(R[s]=x,(3)),Rs);
```

$$p := [-25.0\,\mathrm{LambertW}(3.999600040\times 10^{-12}\,\mathrm{e}^{39.99600040\,U+0.1199880012})-0.0999900\,U+2.999700030, -2.50\,\mathrm{LambertW}(3.996003996\times 10^{-11}\,\mathrm{e}^{39.96003996\,U+1.198801199})-0.09990010\,U+2.997002997, -0.5000000000\,\mathrm{LambertW}(1.990049751\times 10^{-10}\,\mathrm{e}^{39.80099502\,U+5.970149252})-0.09950248800\,U+2.985074626, -0.250\,\mathrm{LambertW}(3.960396040\times 10^{-10}\,\mathrm{e}^{39.60396040\,U+11.88118812})-0.099009901\,U+2.970297030, -0.1250000000\,\mathrm{LambertW}(7.843137255\times 10^{-10}\,\mathrm{e}^{39.21568628\,U+23.52941176})-0.09803921550\,U+2.941176470] \quad \textbf{(5)}$$

```
> plot(p,U=0..0.6,legend=[seq(R[s]=Rs[k],k=1..5)]);
```



c) Die elektrische Leistung ist definiert durch $P = U\,I$. Berechnen Sie für die Solarzelle mit den Parametern aus Teilaufgabe b) und $R_s = 0.05$ die maximal mögliche Leistung.

Hinweise: Schreiben Sie die Leistung P als Funktion von U, d.h. als $P(U)$. Plotten Sie die Funktion $P(U)$. Bestimmen Sie dann die maximal mögliche Leistung $P_{max} = P(U_{max})$. Verwenden Sie fsolve für die Bestimmung der Nullstelle der Ableitung von $P(U)$.

```
> p[3];
```

$$-0.5000000000\,\mathrm{LambertW}(1.990049751\times 10^{-10}\,\mathrm{e}^{39.80099502\,U+5.970149252})-0.09950248800\,U+2.985074626 \quad \textbf{(6)}$$

```
> P:=unapply(U*(6),U);
```

$$P := U \mapsto U \cdot \left(-0.5000000000 \cdot \mathrm{LambertW}\left(1.990049751 \times 10^{-10} \cdot e^{39.80099502 \cdot U + 5.970149252} \right) - 0.09950248800 \cdot U + 2.985074626 \right) \quad (7)$$

```
> plot(P(U),U=0..0.6);
```



Für die maximale Leistung bestimmen wir die Nullstelle der Ableitung von $P(U)$ **.**

```
> U[max]:=fsolve(diff(P(U),U)=0,U);
```

$$U_{max} := 0.4057011910 \quad (8)$$

Die maximal mögliche Leistung ist somit:

```
> P(U[max]);
```

$$1.095065873 \quad (9)$$

Zusatz:

```
> with(plots):
  display([plot(P(U), U = 0 .. 0.6), pointplot([U[max], (9)],
  symbolsize=30)]);
```