

Numerische Differentiation

Es soll die erste Ableitung der Funktion $f := x \mapsto \sin(x) \ln(x)$ an der Stelle $x_0 = 0.5$ mit der zentralen Differenzenformel

```
> Dfz := (f(x[0]+h) - f(x[0]-h)) / (2*h);
```

$$Dfz := \frac{1}{2} \frac{f(x_0 + h) - f(x_0 - h)}{h} \quad (1)$$

und der Differenzenformel

```
> Df3 := 1/6*1/h*(-9*f(x[0]+2*h)+2*f(x[0]+3*h)-11*f(x[0])+18*f(x[0]+h));
```

$$Df3 := \frac{1}{6} \frac{-9f(x_0 + 2h) + 2f(x_0 + 3h) - 11f(x_0) + 18f(x_0 + h)}{h} \quad (2)$$

bei 40 signifikanten Dezimalstellen berechnet werden.

a) Berechnen Sie die logarithmierten absoluten Fehler von Dfz und $Df3$ bei den Schrittweiten $h_i = 10^{-i}$, $i = 1 \dots 10$. Nutzen Sie `seq` für die Generierung der Punktpaare $[i, \log_{10}(|D(f)(x_0) - Dfz|)]$ und $[i, \log_{10}(|D(f)(x_0) - Df3|)]$. Berechnen Sie die linearen Ausgleichsgeraden für diese Punktesfolgen mit `LeastSquares` aus dem Paket `CurveFitting` und ermitteln Sie damit die Fehlerordnungen der Differenzenformeln. Plotten Sie die Geraden in ein Schaubild.


b) Listen Sie die absoluten Fehler für die Schrittweiten $h_i = 10^{-i}$, $i = 1 \dots 20$, auf. Erzeugen Sie eine Tabelle mit Spalten für die Schrittweite h_i und die absoluten Fehler von Dfz und $Df3$. Nutzen Sie `printf` für die formatierte Ausgabe. Interpretieren Sie das Ergebnis.

Lösung a):

```
> f := x -> sin(x)*ln(x);
```

$$f := x \mapsto \sin(x) \ln(x) \quad (3)$$

```
> plot(f(x), x=0..1);
```



```
> x[0]:=0.5;
```

$x_0 := 0.5$

(4)

```
> Digits:=40;
```

$Digits := 40$

(5)

Berechnung der absoluten Fehler:

```
> h:=10.0^(-i):
```

```
  d1:=abs(D(f)(x[0])-Dfz):
```

```
  d2:=abs(D(f)(x[0])-Df3):
```

Punktepaare $[i, \log_{10}(|D(f)(x_0) - Dfz|)]$ und $[i, \log_{10}(|D(f)(x_0) - Df3|)]$:

```
> P1:=seq([i,log10(d1)],i=1..10):
```

```
> P2:=seq([i,log10(d2)],i=1..10):
```

Bestimmung der linearen Ausgleichsgeraden:

```
> with(CurveFitting):
```

```
> E1:=unapply(LeastSquares(P1,n),n);
```

$E1 := n \rightarrow -0.06659653518379777419892186238840070836657$

(6)

$-2.000213878534159259568251370640478589762 n$

```
> E2:=unapply(LeastSquares(P2,n),n);
```

$E2 := n \rightarrow 0.4057122113765992282882990810271145277061$


(7)

$-2.976626384853102257938648870485535959881 n$

Der lineare Koeffizient bestimmt die Fehlerordnung der Differenzenformel, d.h. Dfz hat die Fehlerordnung 2 und $Df3$ die Fehlerordnung 3. Plotten der Ausgleichsgeraden:

```
> p1:=plot([E1(n),P1],n=1..10,style=[line,point],color=[red,red]):
```

```
> p2:=plot([E2(n),P2],n=1..10,style=[line,point],color=[blue,blue])
:
> with(plots):
> display([p1,p2]);
```



Die Steigung der Ausgleichsgeraden gibt die Fehlerordnung der Differenzenformel an.

Lösung b):

Tabelle mit formatierter Ausgabe der Gleitkommazahlen:

```
> seq(sprintf("%10.2e %10.2e %10.2e\n",h,d1,d2),i=1..20);
```

1.00e-01	8.62e-03	1.89e-03
1.00e-02	8.55e-05	3.32e-06
1.00e-03	8.55e-07	3.55e-09
1.00e-04	8.55e-09	3.57e-12
1.00e-05	8.55e-11	3.57e-15
1.00e-06	8.55e-13	3.57e-18
1.00e-07	8.55e-15	3.57e-21
1.00e-08	8.55e-17	3.57e-24
1.00e-09	8.55e-19	3.57e-27
1.00e-10	8.55e-21	4.76e-30
1.00e-11	8.55e-23	2.69e-29
1.00e-12	8.55e-25	6.43e-30
1.00e-13	8.16e-27	2.67e-27
1.00e-14	6.60e-28	1.40e-26
1.00e-15	3.07e-26	6.40e-26

1.00e-16	2.69e-25	6.03e-25
1.00e-17	2.73e-24	1.27e-23
1.00e-18	1.27e-23	9.61e-23
1.00e-19	2.37e-22	2.37e-22
1.00e-20	2.37e-22	1.48e-20

Die Differenzbildung mit anschliessender Division durch sehr kleine Schrittweiten verursacht Auslöschung signifikanter Dezimalstellen bei Gleitpunktoperationen mit endlicher Mantisse (hier 40). Die Genauigkeit nimmt bei der Differenzenformel \mathcal{D}_3 für $i \geq 10$ und bei der Differenzenformel \mathcal{D}_5 für $i \geq 14$ kontinuierlich ab.