```
> restart:
```

Musterlösung Aufgabe 6

Parallele Programmierung mit dem Grid Programming Model

a) Schreiben Sie mit Hilfe des Grid Programming Models (Grid) eine Prozedur `myadd(f,i,n)` zur Addition der Zahlen `f(1) + f(2) + ... + f(n)`. Hierbei sei `f(i)` ein von `i` abhängiger Ausdruck.

Hinweise: Werten Sie die Terme `f(i)` mit evalf aus. Siehe auch `mySeq(f,i,n)` in Kapitel 8.5.1 des Skripts.

```
> with(Grid);
```

[*Barrier*, *Get*, *GetLastResult*, *Interrupt*, *Launch*, *Map*, *MyNode*, *NumNodes*, *Receive*, *Run*, *Send*, *Seq*, *Server*, *Set*, *Setup*, *Status*, *Wait*, *WaitForFirst*] **(1)**

```
> myadd:=proc(f,i,n)
    local thisNode,N,result,div;
    uses Grid;
    #
    thisNode:=MyNode();
    N:=NumNodes();
    div:=floor(n/N);
    #
    result:=add(evalf(eval(f,i=j)),j=thisNode*div+1..(thisNode+1)*
  div):
    #
    if(thisNode>0) then
      Send(0,result);
    else
      result+add(Receive(i),i=1..N-1)+add(evalf(eval(f,i=j)),j=div*
  N+1..n);
    end if:
  #
  end proc:
```

b) Berechnen Sie die Summe $\sum_{i=1}^{n} \frac{1}{i^2}$ mit $n = 10^7$ und messen Sie die benötigte Laufzeit von `myadd` auf 4 und 8 Knoten (Threads). Messen Sie zum Vergleich die Laufzeit, die die Maple-Funktion add benötigt.

```
> start:= time[real]():
  Launch(myadd,1/i^2,i,10^7,numnodes=4);
  time[real]()-start;
```

$$1.644933967$$

$$5.067 \qquad (2)$$

```
> start:= time[real]():
  Launch(myadd,1/i^2,i,10^7,numnodes=8);
  time[real]()-start;
```

$$1.644933967$$

$$4.392 \qquad (3)$$

```
> start:= time[real]():
  result:=add(evalf(1/i^2),i=1..10^7);
  time[real]()-start;
```

$$result := 1.644933967$$

$$17.635 \qquad (4)$$

c) Bestimmen Sie den exakten Grenzwert der Summe $\sum_{i=1}^{\infty} \frac{1}{i^2}$ mit der Prozedur sum.

Wieviel Dezimalstellen Genauigkeit hat die Partilasumme $\sum_{i=1}^{10^7} \frac{1}{i^2}$?

```
> sum(1/i^2,i=1..infinity);
```

$$\frac{\pi^2}{6} \qquad (5)$$

```
> evalf((5));
```

$$1.644934068 \qquad (6)$$

```
> result - (6) ;
```

$$-1.01\ 10^{-7} \qquad (7)$$

Partialsumme und exaktes Ergebnis stimmen in den ersten 5 Dezimalstellen bzw. in den ersten 6 signifikanten Stellen überein.