```
> restart:
```

Aufgabe 8

Unbestimmte Integration rationaler Funktionen, Bernoulli-Darstellung

Gegeben sei die rationale Funktion $r(x) = \frac{x^{10} + 2x^7 - 5x^5 - x + 1}{x^8 + x^7 - 2x^6 - x^5 + 2x^4 - x^3 - 2x^2 + x + 1}$ **.**

a) Berechnen Sie das unbestimmte Integral in der Bernoulli-Darstellung, d.h. zeigen Sie, dass sich die Stammfunktion in der Form

$$\int r(x)\,dx = \int P(x)\,dx + \sum_{i=1}^{M} \frac{p_i(x)}{(x-a_i)^{\nu_i - 1}} + \sum_{i=1}^{M} A_i \cdot \log(x - a_i)$$

schreiben lässt. Hierbei sind die $a_i \in \mathbb{C}, i = 1, ..., M,$ **die verschiedenen Nullstellen des Nennerpolynoms mit den jeweiligen Vielfachheiten** ν_i **und** $P(x), p_i(x) \in \mathbb{C}[x]$ **Polynome mit** $deg(P(x)) \le deg(r(x)) = deg(p(x)) - deg(q(x))$ **sowie** $deg(p_i(x)) < \nu_i - 1$**.**

Hinweis: Nutzen Sie convert,fullparfrac für die Zerlegung des Nennerpolynoms in Linearfaktoren.

```
> r := (x**10 + 2*x**7 - 5*x**5 - x + 1)/(x**8 + x**7 - 2*x**6 - x*
  *5 + 2*x**4 - x**3 - 2*x**2 + x + 1);
```

$$r := \frac{x^{10} + 2x^7 - 5x^5 - x + 1}{x^8 + x^7 - 2x^6 - x^5 + 2x^4 - x^3 - 2x^2 + x + 1} \qquad \textbf{(1)}$$

```
> f := convert(r, fullparfrac);
```

$$f := x^2 - x + 3 + \left(\sum_{\_\alpha = RootOf(\_Z^2 - \_Z + 1)} \frac{\frac{\_\alpha}{9} + \frac{2}{9}}{x - \_\alpha} \right) - \frac{1}{8(x-1)^2} + \frac{1}{4(x-1)} - \frac{101}{36(x+1)} - \frac{5}{6(x+1)^3} + \frac{1}{2(x+1)^4} + \frac{47}{24(x+1)^2} \qquad \textbf{(2)}$$

```
> bernoulli_integral := int(convert(f, radical), x);
```

$$bernoulli\_integral := 3x + \frac{x^3}{3} + \frac{5\ln(4x^2 - 4x + 4)}{18} - \frac{\sqrt{3}\arctan\left(\frac{(2x-1)\sqrt{3}}{3}\right)}{9} - \frac{x^2}{2} + \frac{1}{8(x-1)} + \frac{\ln(x-1)}{4} - \frac{1}{6(x+1)^3} + \frac{5}{12(x+1)^2} - \frac{47}{24(x+1)} \qquad \textbf{(3)}$$

$$-\frac{101\ln(x+1)}{36}$$

Das ist nicht die Bernoulli-Darstellung !
Vertauschung der Reihenfolge der Befehle int und convert führt zur gewünschten Bernoulli-Darstellung des unbestimmten Intergrals.

```
> bernoulli_integral := convert(int(f,x),radical);
```

$$bernoulli\_integral := 3x+\frac{x^3}{3}-\frac{x^2}{2}+\frac{1}{8(x-1)}+\frac{\ln(x-1)}{4}-\frac{1}{6(x+1)^3}+\frac{5}{12(x+1)^2}-\frac{47}{24(x+1)}-\frac{101\ln(x+1)}{36}+\frac{\ln\left(x-\frac{1}{2}-\frac{I\sqrt{3}}{2}\right)\left(\frac{1}{2}+\frac{I\sqrt{3}}{2}\right)}{9}+\frac{2\ln\left(x-\frac{1}{2}-\frac{I\sqrt{3}}{2}\right)}{9}+\frac{\ln\left(x-\frac{1}{2}+\frac{I\sqrt{3}}{2}\right)\left(\frac{1}{2}-\frac{I\sqrt{3}}{2}\right)}{9}+\frac{2\ln\left(x-\frac{1}{2}+\frac{I\sqrt{3}}{2}\right)}{9} \quad (4)$$

```
> bernoulli_integral := collect(bernoulli_integral,ln);
```

$$bernoulli\_integral := \frac{\ln(x-1)}{4}-\frac{101\ln(x+1)}{36}+\left(\frac{5}{18}+\frac{I\sqrt{3}}{18}\right)\ln\left(x-\frac{1}{2}-\frac{I\sqrt{3}}{2}\right)+\left(\frac{5}{18}-\frac{I\sqrt{3}}{18}\right)\ln\left(x-\frac{1}{2}+\frac{I\sqrt{3}}{2}\right)+3x+\frac{x^3}{3}-\frac{x^2}{2}+\frac{1}{8(x-1)}-\frac{1}{6(x+1)^3}+\frac{5}{12(x+1)^2}-\frac{47}{24(x+1)} \quad (5)$$

b) Zeigen Sie, dass sich die Lösung des Maple-Integrators `int` von der Bernoulli-Darstellung um eine Konstante unterscheidet. Ermitteln Sie diese Konstante und plotten Sie die Differenz der Lösungen.

```
> maple_int := int(r,x);
```

$$maple\_int := \frac{x^3}{3}-\frac{x^2}{2}+3x+\frac{1}{8(x-1)}+\frac{\ln(x-1)}{4}-\frac{1}{6(x+1)^3}+\frac{5}{12(x+1)^2}-\frac{47}{24(x+1)}-\frac{101\ln(x+1)}{36}+\frac{5\ln(x^2-x+1)}{18}-\frac{\sqrt{3}\arctan\left(\frac{(2x-1)\sqrt{3}}{3}\right)}{9} \quad (6)$$

```
> difference := simplify(bernoulli_integral - maple_int);
```

$$difference := -\frac{5\ln(2)}{9}+\frac{5\ln\left(2x-1-I\sqrt{3}\right)}{18}+\frac{I\sqrt{3}\ln\left(2x-1-I\sqrt{3}\right)}{18} \quad (7)$$

$$+\frac{5\ln\left(\mathrm{I}\sqrt{3}+2x-1\right)}{18}-\frac{\mathrm{I}\sqrt{3}\ln\left(\mathrm{I}\sqrt{3}+2x-1\right)}{18}-\frac{5\ln\left(x^2-x+1\right)}{18}$$

$$+\frac{\sqrt{3}\arctan\left(\frac{(2x-1)\sqrt{3}}{3}\right)}{9}$$

```
> plot([bernoulli_integral, maple_int, difference], x=0..2);
```




```
> simplify(diff(difference,x));
```

$$0 \tag{8}$$

verschwindet, also unterscheiden sich die Stammfunktionen nur um eine Konstante.

```
> simplify(subs(x=0,difference)); # Durch Einsetzen eines
  speziellen Wertes erhält man den Wert der Konstanten.
```

$$\frac{\sqrt{3}\,\pi}{18} \tag{9}$$

```
> evalf(%);
```

$$0.3022998942 \tag{10}$$

```
> plot(difference,x=-10..10);
```




```
> plot(difference,x=-10..10,0.3..0.31);  # Mit gröberer Skalierung
  der y-Achse sieht man die Abweichung  nicht.
```