```
> restart:
```

Musterlösung Aufgabe 9

Integration rationaler Funktionen: Logarithmische Stammfunktion

Gegeben sei die rationale Funktion

```
> r:=x->(13*x^4+8*x^5+15*x^3+28*x+10*x^7+13*x^2-2)/(2*x^8+2*x^6+3*x^5+6*x^3-3*x^4+3*x^2+2);
```

$$r := x \mapsto \frac{13\,x^4 + 8\,x^5 + 15\,x^3 + 28\,x + 10\,x^7 + 13\,x^2 - 2}{2\,x^8 + 2\,x^6 + 3\,x^5 + 6\,x^3 - 3\,x^4 + 3\,x^2 + 2} \qquad (1)$$

Die unbestimmte Integration mittels `int` liefert die folgende Stammfunktion bestehend aus zwei logarithmischen Anteilen

```
> int(r(x),x);
```

$$\left(\sum_{\_R = RootOf(\_Z^3 - \_Z^2 + 1)} \frac{\left(-3\,\_R^2 + 5\,\_R - 1\right)\ln(x - \_R)}{3\,\_R^2 - 2\,\_R}\right) + \ln\left(x^2 + 2\right) + 2\ln\left(2\,x^3 + 2\,x^2 + 1\right) \qquad (2)$$

a) Zeigen Sie, dass die Terme außerhalb des Summenzeichens (unabhängig vom Laufindex $\_R$) die Form

$$\int \frac{c(x)}{d(x)} = \sum_{k=1}^{n} z_k \cdot \ln\left(v_k(x)\right)$$

haben (Rothstein-Trager). Hierbei ist $\frac{c(x)}{d(x)}$ ein Teil des Integranden $r(x)$, die z_k sind die verschiedenen Lösungen der Resultante (resultant) der Polynome $c(x) - z\,d'(x)$ und $d(x)$ und $v_k(x)$ ist die Funktion $gcd\left(c(x) - z_k d'(x), d(x)\right)$.

Hinweis: Nutzen Sie für die Partialbruchzerlegung von $r(x)$ convert,parfrac und für die Bestimmung der Funktionsanteile applyrule oder select.

Lösung a)

Wir berechnen zunächst die Partialbruchzerlegung von $r(x)$:

```
> p:=convert(r(x),parfrac);
```

(3)

$$p := \frac{-3x^2+5x-1}{x^3-x^2+1} + \frac{2x}{x^2+2} + \frac{4x(3x+2)}{2x^3+2x^2+1} \qquad (3)$$

Die Terme mit den Nennerpolynomen $2x^3+2x^2+1$ und x^2+2 gehören zum logarithmischen Teil der Stammfunktion außerhalb des Summenzeichens. Wir extrahieren die Anteile $p1$ und $p2$ mittels applyrule wie folgt:

```
> p1:=applyrule([-1+5*x-3*x^2=0],p);
```

$$p1 := \frac{2x}{x^2+2} + \frac{4x(3x+2)}{2x^3+2x^2+1} \qquad (4)$$

```
> p2:=applyrule([4*x*(2+3*x)=0,2*x=0],p);
```

$$p2 := \frac{-3x^2+5x-1}{x^3-x^2+1} \qquad (5)$$

Alternative direkte Bestimmung (mittels select und has):

```
> select(not has,(2),sum);
```

$$\ln(x^2+2) + 2\ln(2x^3+2x^2+1) \qquad (6)$$

Die option `RootOf` wäre ebenso möglich:

```
> select(not has,(2),RootOf);
```

$$\ln(x^2+2) + 2\ln(2x^3+2x^2+1) \qquad (7)$$

```
> p1:=diff((6),x);
```

$$p1 := \frac{2x}{x^2+2} + \frac{2(6x^2+4x)}{2x^3+2x^2+1} \qquad (8)$$

```
> select(has,(2),sum);
```

$$\sum_{\_R=RootOf(\_Z^3-\_Z^2+1)} \frac{(-3\_R^2+5\_R-1)\ln(x-\_R)}{3\_R^2-2\_R} \qquad (9)$$

```
> p2:=diff((9),x);
```

$$p2 := \frac{-3x^2+5x-1}{x^3-x^2+1} \qquad (10)$$

Die Summanden $p1$ und $p2$ werden jetzt separat weiterbehandelt. Zunächst Bestimmung von $\frac{c(x)}{d(x)}$:

```
> d:=expand(denom(p1));
```

$$d := 2x^5+2x^4+4x^3+5x^2+2 \qquad (11)$$

Normierung von $d(x)$:

```
> l:=lcoeff(d);
```

$$l := 2 \qquad (12)$$

```
> d:=unapply(d/l,x);
```

$$d := x \mapsto x^5 + x^4 + 2x^3 + \frac{5}{2}x^2 + 1 \tag{13}$$

```
> c:=unapply(expand(numer(p1))/l,x);
```

$$c := x \mapsto 8x^4 + 6x^3 + 12x^2 + 9x \tag{14}$$

Berechnung der Resultante von $c(x) - z\,d'(x)$ und $d(x)$:

```
> res:=resultant(c(x)-z*diff(d(x),x),d(x),x);
```

$$res := -\frac{58835}{8}z^5 + 58835z^4 - \frac{1470875}{8}z^3 + \frac{1117865}{4}z^2 - \frac{411845}{2}z + 58835 \tag{15}$$

Bemerkung: Ohne Normierung unterscheidet sich die Resultante durch einen konstanten Faktor (512), der für die Nullstellen des Polynoms res keine Rolle spielt.

Es gibt zwei Nullstellen mit den Vielfachheiten 2 und 3.

```
> solve(res,z);
```

$$1, 1, 2, 2, 2 \tag{16}$$

```
> z:=op({(16)});
```

$$z := 1, 2 \tag{17}$$

Berechnung der $v_k(x)$ für z_1 und z_2:

```
> v[1]:=unapply(gcd(c(x)-z[1]*diff(d(x),x),d(x)),x);
```

$$v_1 := x \mapsto x^2 + 2 \tag{18}$$

```
> v[2]:=unapply(gcd(c(x)-z[2]*diff(d(x),x),d(x)),x);
```

$$v_2 := x \mapsto x^3 + x^2 + \frac{1}{2} \tag{19}$$

Nach dem Satz von Rothstein-Trager folgt damit

```
> Int(p1,x)=add(z[i]*ln(v[i](x)),i=1..2);
```

$$\int \left(\frac{2x}{x^2+2} + \frac{2(6x^2+4x)}{2x^3+2x^2+1} \right) \mathrm{d}x = \ln(x^2+2) + 2\ln\left(x^3+x^2+\frac{1}{2}\right) \tag{20}$$

Probe:

```
> Int(p1,x)=int(p1,x);
```

$$\int \left(\frac{2x}{x^2+2} + \frac{2(6x^2+4x)}{2x^3+2x^2+1} \right) \mathrm{d}x = \ln(x^2+2) + 2\ln(2x^3+2x^2+1) \tag{21}$$

b) Zeigen Sie, dass die Summe über $\_R$ (zweiter Anteil der logarithmischen Stammfunktion) die allgemeine Form

$$\sum_{\{a \mid g(a)=0\}} \frac{f(a)}{g'(a)} \cdot \ln(x-a)$$

hat. Der Summationsindex a läuft über alle Nullstellen des Nennerpolynoms $g(x)$.

Lösung b)

Der Summand $p2$ hat die Stammfunktion

```
> int(p2,x);
```

$$\sum_{\_R=RootOf(\_Z^3-\_Z^2+1)} \frac{(-3\ \_R^2+5\ \_R-1)\ \ln(x-\_R)}{3\ \_R^2-2\ \_R} \tag{22}$$

Wir definieren

```
> f:=unapply(numer(p2),x);
```

$$f := x \mapsto -3\,x^2+5\,x-1 \tag{23}$$

```
> g:=unapply(denom(p2),x);
```

$$g := x \mapsto x^3-x^2+1 \tag{24}$$

Mit

```
> c:=unapply(f(x)/diff(g(x),x),x);
```

$$c := x \mapsto \frac{-3\,x^2+5\,x-1}{3\,x^2-2\,x} \tag{25}$$

folgt die Darstellungsform unter dem Summenzeichen.

```
> (22) = sum(f(a)/diff(g(a),a)*ln(x-a),a=RootOf(g(x)=0));
```

$$\sum_{\_R=RootOf(\_Z^3-\_Z^2+1)} \frac{(-3\ \_R^2+5\ \_R-1)\ \ln(x-\_R)}{3\ \_R^2-2\ \_R} = \sum_{a=RootOf(\_Z^3-\_Z^2+1)} \frac{(-3\,a^2+5\,a-1)\ \ln(x-a)}{3\,a^2-2\,a} \tag{26}$$

<u>Zusatz: Explizite Stammfunktion mittels Rothstein-Trager</u>

```
> RothsteinTrager:=proc(r::ratpoly,x::symbol)
     local c,d,l,z,res,sol,zlist,vlist:
     uses ListTools:
     d:=denom(r):
     l:=lcoeff(d):
```

```
    c:=numer(r)/l:
    d:=d/l:
    res:=resultant(c-z*diff(d,x),d,x):
    zlist:=convert({solve(res=0,z)},list):
    vlist:=map(z->gcd(c-z*diff(d,x),d),zlist):
    DotProduct(zlist,map(z->ln(z),vlist)):
  end proc:
```

```
> RothsteinTrager(r(x),x);
```

$$\ln\left(x^2+2\right)+2\ln\left(x^3+x^2+\frac{1}{2}\right)+\left(-\frac{\left(1428300+171948\sqrt{69}\right)^{1/3}}{138}+\frac{2}{\left(1428300+171948\sqrt{69}\right)^{1/3}}-1\right)\ln\left(-\frac{\left(1428300+171948\sqrt{69}\right)^{2/3}\sqrt{69}}{552}+\frac{25\left(1428300+171948\sqrt{69}\right)^{1/3}\sqrt{69}}{828}+\frac{25\left(1428300+171948\sqrt{69}\right)^{2/3}}{1656}+x-\frac{\left(1428300+171948\sqrt{69}\right)^{1/3}}{4}-\frac{1}{3}\right)+\left(\frac{\left(1428300+171948\sqrt{69}\right)^{1/3}}{276}-\frac{1}{\left(1428300+171948\sqrt{69}\right)^{1/3}}-1-\frac{I\sqrt{3}\left(-\frac{\left(1428300+171948\sqrt{69}\right)^{1/3}}{138}-\frac{2}{\left(1428300+171948\sqrt{69}\right)^{1/3}}\right)}{2}\right)\ln\left(\frac{25\,I\sqrt{3}\left(1428300+171948\sqrt{23}\sqrt{3}\right)^{2/3}}{3312}+\frac{\left(1428300+171948\sqrt{23}\sqrt{3}\right)^{2/3}\sqrt{23}\sqrt{3}}{1104}-\frac{I\left(1428300+171948\sqrt{23}\sqrt{3}\right)^{2/3}\sqrt{23}}{368}+\frac{I\sqrt{3}\left(1428300+171948\sqrt{23}\sqrt{3}\right)^{1/3}}{8}\right.\tag{27}$$

$$- \frac{25 \sqrt{23} \sqrt{3} \left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{1/3}}{1656}$$

$$- \frac{25 \mathrm{I} \left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{1/3} \sqrt{23}}{552}$$

$$- \frac{25 \left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{2/3}}{3312} + x + \frac{\left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{1/3}}{8}$$

$$- \frac{1}{3} \Bigg) + \left(\frac{\left(1428300 + 171948 \sqrt{69}\right)^{1/3}}{276} - \frac{1}{\left(1428300 + 171948 \sqrt{69}\right)^{1/3}} - 1 \right.$$

$$\left. + \frac{\mathrm{I} \sqrt{3} \left(- \frac{\left(1428300 + 171948 \sqrt{69}\right)^{1/3}}{138} - \frac{2}{\left(1428300 + 171948 \sqrt{69}\right)^{1/3}} \right)}{2} \right) \ln\Bigg($$

$$- \frac{25 \mathrm{I} \sqrt{3} \left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{2/3}}{3312}$$

$$+ \frac{\left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{2/3} \sqrt{23} \sqrt{3}}{1104}$$

$$+ \frac{\mathrm{I} \left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{2/3} \sqrt{23}}{368}$$

$$- \frac{\mathrm{I} \sqrt{3} \left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{1/3}}{8}$$

$$- \frac{25 \sqrt{23} \sqrt{3} \left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{1/3}}{1656}$$

$$+ \frac{25 \mathrm{I} \left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{1/3} \sqrt{23}}{552}$$

$$- \frac{25 \left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{2/3}}{3312} + x + \frac{\left(1428300 + 171948 \sqrt{23} \sqrt{3}\right)^{1/3}}{8}$$

$$- \frac{1}{3} \Bigg)$$

Implizite und explizite Stammfunktion sind bis auf eine Konstante identisch:

```
> simplify(diff((2)-(27),x));
```

$$0 \tag{28}$$