```
> restart:
> plots:-setcolors("Niagara"): # Das ist das Default-Farbschema, das ich nicht verwende;
                               # Mir sind die Farben zu ähnlich, kaum zu unterscheiden.
```

Aufgabe 1

Ungleichung mit komplexen Zahlen, Ellipsengleichung

Für welche komplexen Zahlen $z \in \mathbb{C}$ ist die folgende Ungleichung erfüllt?

```
> abs(z-1)<abs(5/2*z-conjugate(z)-I);
```

$$|z-1| < \left|\frac{5\,z}{2} - \bar{z} - \mathrm{I}\right| \qquad \textbf{(1)}$$

a) Formulieren Sie die Ungleichung in der Form $f(x,y) > 0$. Die Argumente der Funktion f seien hierbei der Realteil x und der Imaginärteil y der komplexen Zahl $z = x + I y$.

Hinweis: Nutzen Sie evalc für die Auswertung symbolischer Ausdrücke mit komplexen Zahlen und map für Äquivalenzumformungen, z.B. `map(t->t^2,Ungleichung)` . Linke und rechte Seite einer Gleichung bzw. Ungleichung erhalten Sie mit lhs bzw. rhs.

```
> z:=x+I*y;
  f(z):=evalc(abs(z-1)<abs(5/2*z-conjugate(z)-I));
  f(z) := evalc((1));                                # besser;
  f(x,y):=rhs(map(t->t-lhs(f(z)), f(z)));            # Vorsicht:
                                                     # f(x,y) ist keine Funktion !
                                                     # f(x,y) ist ein Ausdruck, der den Namen f(x,y) hat.
                                                     #
  f(x,y)>0;
```

$$z := x + \mathrm{I}\,y$$

$$f(x+\mathrm{I}\,y) := \sqrt{y^2 + (x-1)^2} < \frac{\sqrt{9\,x^2 + 49\,y^2 - 28\,y + 4}}{2}$$

$$f(x+\mathrm{I}\,y) := \sqrt{y^2 + (x-1)^2} < \frac{\sqrt{9\,x^2 + 49\,y^2 - 28\,y + 4}}{2}$$

$$f(x,y) := \frac{\sqrt{9\,x^2 + 49\,y^2 - 28\,y + 4}}{2} - \sqrt{y^2 + (x-1)^2}$$

$$0 < \frac{\sqrt{9x^2+49y^2-28y+4}}{2} - \sqrt{y^2+(x-1)^2} \qquad \textbf{(2)}$$

$f(x,y)$ **lässt sich so nicht vereinfachen. Beachte den Hinweis in der Aufgabenstellung! Bevor Du alles auf eine Seite bringst, ist es sinnvoll zu quadrieren!**

Äquivalenzumformungen:

```
> evalc((1));
  map(t->t^2,%);
  rhs(%)-lhs(%)>0;
  simplify(%);
  %*4/5;
```

$$\sqrt{y^2+(x-1)^2} < \frac{\sqrt{9x^2+49y^2-28y+4}}{2}$$

$$y^2+(x-1)^2 < \frac{49}{4}y^2-7y+1+\frac{9}{4}x^2$$

$$0 < \frac{45y^2}{4}-7y+1+\frac{9x^2}{4}-(x-1)^2$$

$$0 < \frac{45}{4}y^2-7y+\frac{5}{4}x^2+2x$$

$$0 < 9y^2-\frac{28}{5}y+x^2+\frac{8}{5}x \qquad \textbf{(3)}$$

```
> sort(rhs((3)));
  f:= unapply(%,[x,y]);        # Funktion f(x,y);
```

$$x^2+9y^2+\frac{8}{5}x-\frac{28}{5}y$$

$$f := (x,y) \mapsto x^2+9\cdot y^2+\frac{8}{5}\cdot x-\frac{28}{5}\cdot y \qquad \textbf{(4)}$$

b) Zeichnen Sie die nichtlineare Ungleichung $f(x,y) > 0$.

Hinweis: Nutzen Sie die Funktion implicitplot aus dem Paket plots und färben Sie die Fläche mit $f(x,y) > 0$ **ein (Option `filled=true, grid=[1000,1000]`).**

```
> with(plots, implicitplot):
  implicitplot(f(x,y)>0, filled=true, grid=[1000,1000], scaling=
  constrained);
```



c) Beschreiben Sie verbal die Lösung der Ungleichung mit Hilfe der Ellipsengleichung $f(x, y) = 0$.

```
> # Die Randpunkte erfüllen die Lösung der Ungleichung und bilden daher eine
  Ellipse.
```

Das ist falsch. Der rote Bereich ist doch die Lösung!

Die komplexe Ungleichung wird für alle $z = x + I \cdot y \in \mathbb{C}$ außerhalb der Ellipsenlinie

```
> f(x,y)=0;
```

$$x^2 + 9\,y^2 + \frac{8}{5}\,x - \frac{28}{5}\,y = 0 \tag{5}$$

erfüllt.

d) Ermitteln Sie die Mittelpunktsform der Ellipsengleichung algebraisch durch quadratische Ergänzung der Polynomfunktion zweiten Grades $f(x, y)$.

Hinweis: Verwenden Sie CompleteSquare aus dem Paket Student[Precalculus].

```
> with(Student[Precalculus]):
  CompleteSquare(f(x,y), x, y);
```

$$9\left(y - \frac{14}{45}\right)^2 + \left(x + \frac{4}{5}\right)^2 - \frac{68}{45} \tag{6}$$

```
> expand(%)=0;          # Ellipsengleichung
```

$$x^2 + 9\,y^2 + \frac{8}{5}\,x - \frac{28}{5}\,y = 0 \tag{7}$$

Mittelpunktsform:

```
> (6)*45/68+1=1;
```

$$\frac{405\left(y - \frac{14}{45}\right)^2}{68} + \frac{45\left(x + \frac{4}{5}\right)^2}{68} = 1 \tag{8}$$