

> restart;

Musterlösung Aufgabe 1

Ungleichung mit komplexen Zahlen, Ellipsengleichung

Für welche komplexen Zahlen $z \in \mathbb{C}$ ist die folgende Ungleichung erfüllt?

> abs(z-1)<abs(5/2*z-conjugate(z)-I);

$$|z-1| < \left| \frac{5z}{2} - \bar{z} - I \right| \quad (1)$$

a) Formulieren Sie die Ungleichung in der Form $f(x, y) > 0$. Die Argumente der Funktion f seien hierbei der Realteil x und der Imaginärteil y der komplexen Zahl $z = x + Iy$.

Hinweis: Nutzen Sie [evalc](#) für die Auswertung symbolischer Ausdrücke mit komplexen Zahlen und [map](#) für Äquivalenzumformungen, z.B. `map(t->t^2, Ungleichung)`. Linke und rechte Seite einer Gleichung bzw. Ungleichung erhalten Sie mit [lhs](#) bzw. [rhs](#).

> z:=x+I*y;

$$z := x + Iy \quad (2)$$

> evalc((1));

$$\sqrt{(x-1)^2 + y^2} < \frac{\sqrt{9x^2 + 49y^2 - 28y + 4}}{2} \quad (3)$$

Äquivalenzumformungen:

> map(t->t^2,%);

$$(x-1)^2 + y^2 < \frac{9}{4}x^2 + \frac{49}{4}y^2 - 7y + 1 \quad (4)$$

> expand((rhs-lhs)(%))>0;

$$0 < \frac{5}{4}x^2 + \frac{45}{4}y^2 - 7y + 2x \quad (5)$$

> map(t->t*4/5,%);

$$0 < x^2 + 9y^2 - \frac{28}{5}y + \frac{8}{5}x \quad (6)$$


> f:=unapply(rhs(%),x,y);

$$f := (x, y) \mapsto x^2 + 9y^2 - \frac{28}{5}y + \frac{8}{5}x \quad (7)$$

b) Zeichnen Sie die nichtlineare Ungleichung $f(x, y) > 0$.

Hinweis: Nutzen Sie die Funktion [implicitplot](#) aus dem Paket [plots](#) und färben Sie die Fläche mit $f(x, y) > 0$ ein (Option `filled=true`, `grid=[1000,1000]`).

```
> with(plots):
> p1:=implicitplot(f(x,y)>0,x=-3..1,y=-1..2,scaling=constrained,
  grid=[1000,1000],filled=true,coloring=[blue]):
> p1;
```



c) Beschreiben Sie verbal die Lösung der Ungleichung mit Hilfe der Ellipsengleichung $f(x, y) = 0$.

Die komplexe Ungleichung wird für alle $z = x + I \cdot y \in \mathbb{C}$ außerhalb der Ellipsenlinie

```
> f(x,y)=0;
```

$$x^2 + 9y^2 - \frac{28}{5}y + \frac{8}{5}x = 0 \quad (8)$$

erfüllt.

d) Ermitteln Sie die Mittelpunktsform der Ellipsengleichung algebraisch durch quadratische Ergänzung der Polynomfunktion zweiten Grades $f(x, y)$.

Hinweis: Verwenden Sie [CompleteSquare](#) aus dem Paket [Student\[Precalculus\]](#).

```
> with(Student[Precalculus]);
[CenterOfMass, CompleteSquare, CompositionPlot, CompositionTutor, ConicsTutor, Distance,
  FunctionSlopePlot, FunctionSlopeTutor, LimitPlot, LimitTutor, Line, LineTutor,
  LinearInequalitiesTutor, Midpoint, PolynomialTutor, RationalFunctionPlot,
  RationalFunctionTutor, Slope, StandardFunctionsTutor]
```

```
> CompleteSquare(f(x,y),[x,y]);
```

$$9 \left(y - \frac{14}{45} \right)^2 + \left(x + \frac{4}{5} \right)^2 - \frac{68}{45} \quad (10)$$

```
> expand(%);
```

$$x^2 + 9y^2 - \frac{28}{5}y + \frac{8}{5}x \quad (11)$$

Mittelpunktsform:

> (10)*45/68+1=1;

$$\frac{405 \left(y - \frac{14}{45} \right)^2}{68} + \frac{45 \left(x + \frac{4}{5} \right)^2}{68} = 1$$

(12)